NAG Library Routine Document

F03AEF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F03AEF computes a Cholesky factorization of a real symmetric positive definite matrix, and evaluates the determinant.

2 Specification

```
SUBROUTINE F03AEF (N, A, LDA, P, D1, ID, IFAIL)

INTEGER            N, LDA, ID, IFAIL
REAL (KIND=nag_wp) A(LDA,*), P(N), D1
```

3 Description

F03AEF computes the Cholesky factorization of a real symmetric positive definite matrix $A = LL^{\mathrm{T}}$ where L is lower triangular. The determinant is the product of the squares of the diagonal elements of L.

4 References

Wilkinson J H and Reinsch C (1971) *Handbook for Automatic Computation II, Linear Algebra* Springer–Verlag

5 Parameters

1: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: $\mathrm{N} \geq 0$.

2: A(LDA,$*$) – REAL (KIND=nag_wp) array *Input/Output*

Note: the second dimension of the array A must be at least $\max(1, \mathrm{N})$.

On entry: the upper triangle of the n by n positive definite symmetric matrix A. The elements of the array below the diagonal need not be set.

On exit: the subdiagonal elements of the lower triangular matrix L. The upper triangle of A is unchanged.

3: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F03AEF is called.

Constraint: $\mathrm{LDA} \geq \max(1, \mathrm{N})$.

4: P(N) – REAL (KIND=nag_wp) array *Output*

On exit: the reciprocals of the diagonal elements of L.

5: D1 – REAL (KIND=nag_wp) *Output*
6: ID – INTEGER *Output*

On exit: the determinant of A is given by D1 $\times$ 2.0^{ID}. It is given in this form to avoid overflow or underflow.

7: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL $= 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

The matrix A is not positive definite, possibly due to rounding errors. The factorization could not be completed. D1 and ID are set to zero.

IFAIL $= 2$

On entry, N < 0,
or LDA $< \max(1, \mathrm{N})$.

7 Accuracy

The accuracy of the determinant depends on the conditioning of the original matrix. For a detailed error analysis see page 25 of Wilkinson and Reinsch (1971).

8 Further Comments

The time taken by F03AEF is approximately proportional to n^3.

9 Example

This example computes a Cholesky factorization and calculate the determinant of the real symmetric positive definite matrix

$$\begin{pmatrix} 6 & 7 & 6 & 5 \\ 7 & 11 & 8 & 7 \\ 6 & 8 & 11 & 9 \\ 5 & 7 & 9 & 11 \end{pmatrix}.$$

9.1 Program Text

```
    Program f03aefe

!     F03AEF Example Program Text

!     Mark 24 Release. NAG Copyright 2012.

!     .. Use Statements ..
      Use nag_library, Only: f03aef, nag_wp
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)               :: d1
      Integer                          :: i, id, ifail, lda, n
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: a(:,:), p(:)
!     .. Executable Statements ..
      Write (nout,*) 'F03AEF Example Program Results'

!     Skip heading in data file
      Read (nin,*)

      Read (nin,*) n
      lda = n
      Allocate (a(lda,n),p(n))

      Read (nin,*)(a(i,1:n),i=1,n)

      ifail = 0
      Call f03aef(n,a,lda,p,d1,id,ifail)

      Write (nout,*)
      Write (nout,*) 'Array A after factorization'

      Do i = 1, n
        Write (nout,99999) a(i,1:n)
      End Do

      Write (nout,*)
      Write (nout,*) 'Array P'
      Write (nout,99999) p(1:n)
      Write (nout,*)
      Write (nout,99998) 'D1 = ', d1, '     ID = ', id
      Write (nout,*)
      Write (nout,99998) 'Value of determinant = ', d1*2.0E0_nag_wp**id

99999 Format (1X,8F9.4)
99998 Format (1X,A,F9.4,A,I2)
    End Program f03aefe
```

9.2 Program Data

```
F03AEF Example Program Data
  4
    6    7    6    5
    7   11    8    7
    6    8   11    9
    5    7    9   11
```

9.3 Program Results

```
 F03AEF Example Program Results

 Array A after factorization
    6.0000   7.0000   6.0000   5.0000
    2.8577  11.0000   8.0000   7.0000
    2.4495   0.5941  11.0000   9.0000
```

```
   2.0412   0.6931   1.6645  11.0000

 Array P
   0.4082   0.5941   0.4639   0.5283

 D1 =    0.0691     ID = 12

 Value of determinant =  283.0000
```